In this Maple file we compute the Hamiltonian evolutions of the Darboux coordinates (q,p) for the P1 case. We also split the deformation space into trivial and non-trivial directions.

The deformation operator is \hbar (alpha15\partial_{t_{\infty^{(1)},5} +alpha14\partial_{t_{\infty^{(1)},4}+alpha13*\partial_{t_{\infty^{(1)},3} +alpha12*\partial_{t_{\infty^{(1)},2}+alpha11*\partial_{t_{\infty^{(1)},1}))

```
> restart:
  rinfty:=3:
  with(LinearAlgebra):
  tinfty25:=-tinfty15;
  tinfty23:=-tinfty13;
  tinfty21:=-tinfty11;
  tinfty20:=-tinfty10;
  tinfty24:=tinfty14;
  tinfty22:=tinfty12;
  tinfty10:=0;
  Pinfty11:=-(tinfty14+tinfty24)/2;
  Pinfty01:=-(tinfty12+tinfty22)/2;
  Pinfty32 := -(1/4)*tinfty15^2;
  Pinfty22 := -(1/2)*tinfty15*tinfty13+(1/4)*tinfty14^2;
  Pinfty12 := -(1/2)*tinfty15*tinfty11+(1/2)*tinfty14*tinfty12-
  (1/4)*tinfty13^2;
  P1:=x-> Pinfty01+Pinfty11*x;
  P2:=x-> Pinfty02+Pinfty12*x+Pinfty22*x^2+Pinfty32*x^3;

  c1Asymp := (6*alpha15*tinfty12*tinfty15-6*alpha15*tinfty13*
  tinfty14+10*alpha13*tinfty14*tinfty15-15*alpha12*tinfty15^2)/(30*
  tinfty15^2):
  c2Asymp :=(4*alpha15*tinfty14-5*alpha14*tinfty15)/(20*tinfty15):
  muAsymp := -(2*(3*alpha15*tinfty11*tinfty15-3*alpha15*
  tinfty13^2+5*alpha13*tinfty13*tinfty15-15*alpha11*tinfty15^2))/
  (15*tinfty15^3):
  nuAsymp:= -(2*(3*alpha15*tinfty13-5*alpha13*tinfty15))/(15*
  tinfty15^2):

  dP1dlambda:=unapply(diff(P1(lambda),lambda),lambda):
  dP2dlambda:=unapply(diff(P2(lambda),lambda),lambda):
  L:=Matrix(2,2,0):
  L[1,1]:=0:
  L[1,2]:=1:
  L[2,1]:=-P2(lambda)+Pinfty02 +C - p*h/(lambda-q):
  L[2,2]:= P1(lambda)  +h/(lambda-q) :

  A:=Matrix(2,2,0):
```

```
A[1,1]:=c2*lambda^2+c1*lambda+c0+ rho/(lambda-q):
A[1,2]:=2*alpha15/5/tinfty15*lambda+nu+ mu/(lambda-q):
A[2,1]:=AA21(lambda):
A[2,2]:=AA22(lambda):
dAdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dAdlambda[i,j]:=diff(A
[i,j],lambda):  od: od:

L;
A;

J:=Matrix(2,2,0):
J[1,1]:=1:
J[1,2]:=0:
J[2,1]:=-p/(lambda-q):
J[2,2]:=1/(lambda-q):
dJdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dJdlambda[i,j]:=diff(J
[i,j],lambda): od: od:
J:

CurlyLJ:=Matrix(2,2,0):
CurlyLJ[1,1]:=0:
CurlyLJ[1,2]:=0:
CurlyLJ[2,2]:=diff(J[2,2],q)*CurlyLq+diff(J[2,2],p)*CurlyLp+h*
diff(J[2,2],t):
CurlyLJ[2,1]:=diff(J[2,1],q)*CurlyLq+diff(J[2,1],p)*CurlyLp+h*
diff(J[2,1],t):
CurlyLJ:



Lnew:=simplify(Multiply(Multiply(J,L),J^(-1))+h*Multiply
(dJdlambda,J^(-1))):
Anew:=simplify(Multiply(Multiply(J,A),J^(-1))+Multiply(CurlyLJ,J^
(-1))):
```

$$tinfty25 := -tinfty15$$
$$tinfty23 := -tinfty13$$
$$tinfty21 := -tinfty11$$
$$tinfty20 := -tinfty10$$

$$tinfty24 := tinfty14$$
$$tinfty22 := tinfty12$$
$$tinfty10 := 0$$
$$Pinfty11 := -tinfty14$$
$$Pinfty01 := -tinfty12$$
$$Pinfty32 := -\frac{tinfty15^2}{4}$$
$$Pinfty22 := -\frac{tinfty15\ tinfty13}{2} + \frac{tinfty14^2}{4}$$
$$Pinfty12 := -\frac{tinfty15\ tinfty11}{2} + \frac{tinfty14\ tinfty12}{2} - \frac{tinfty13^2}{4}$$
$$P1 := x \mapsto Pinfty01 + Pinfty11\, x$$
$$P2 := x \mapsto Pinfty02 + Pinfty12\, x + Pinfty22\, x^2 + Pinfty32\, x^3$$

$$\left[\left[0,\ 1\right],\ \left[-\left(-\frac{tinfty15\ tinfty11}{2} + \frac{tinfty14\ tinfty12}{2} - \frac{tinfty13^2}{4}\right)\lambda - \left(-\frac{tinfty15\ tinfty13}{2} + \frac{tinfty14^2}{4}\right)\lambda^2 + \frac{tinfty15^2\,\lambda^3}{4} + C - \frac{p\,h}{\lambda - q},\ -tinfty14\,\lambda - tinfty12 + \frac{h}{\lambda - q}\right],\ \begin{bmatrix} c2\,\lambda^2 + c1\,\lambda + c0 + \frac{\rho}{\lambda - q} & \frac{2\,\alpha 15\,\lambda}{5\ tinfty15} + \nu + \frac{\mu}{\lambda - q} \\ AA21(\lambda) & AA22(\lambda) \end{bmatrix}\right] \quad (1)$$

Using the compatibility equation \mathcal{L}[L]=h\partial_x A+[A,L] we can obtain A[2,1] et A[2,2] from the trivial first line of L(\lambda)

```
> CurlyL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):
  Entry11:=CurlyL[1,1]:
  Entry12:=CurlyL[1,2]:

  AA21:=unapply(solve(Entry11=0,AA21(lambda)),lambda):
  AA21bis:=h*dAdlambda[1,1]+A[1,2]*L[2,1]:

  simplify(AA21(lambda)-AA21bis);
  AA22:=unapply(solve(Entry12=0,AA22(lambda)),lambda):
  AA22bis:=h*dAdlambda[1,2]+A[1,1]+A[1,2]*L[2,2]:

  simplify(AA22(lambda)-AA22bis);
  simplify(Entry11);
  simplify(Entry12);
  CurlyL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):
```

$$0$$
$$0$$
$$0 \quad (2)$$

$$0 \quad \textbf{(2)}$$

The next step is to use the compatibility equation on L[2,2] et L[2,1] to get the evolutions of the Darboux coordinates:

```
> Entry22:=simplify(CurlyL[2,2]):
  Entry22TermLambdaMinusqCube:=factor(residue(Entry22*(lambda-q)^2,
  lambda=q));
  Entry22TermLambdaMinusqCarre:=factor(residue(Entry22*(lambda-q),
  lambda=q));
  Entry22TermLambdaMinusq:=factor(residue(Entry22,lambda=q));

  Entry22TermLambdaInfty4:=factor(-residue(Entry22/lambda^5,lambda=
  infinity));
  Entry22TermLambdaInfty3:=factor(-residue(Entry22/lambda^4,lambda=
  infinity));
  Entry22TermLambdaInfty2:=factor(-residue(Entry22/lambda^3,lambda=
  infinity));
  Entry22TermLambdaInfty1:=factor(-residue(Entry22/lambda^2,lambda=
  infinity));
  Entry22TermLambdaInfty0:=factor(-residue(Entry22/lambda,lambda=
  infinity));

  simplify( Entry22-(Entry22TermLambdaMinusqCarre/(lambda-q)^2+
  Entry22TermLambdaMinusq/(lambda-q)
  +Entry22TermLambdaInfty0+Entry22TermLambdaInfty1*lambda+
  Entry22TermLambdaInfty2*lambda^2+Entry22TermLambdaInfty3*
  lambda^3+Entry22TermLambdaInfty4*lambda^4) );
  L[2,2];
```

$$Entry22TermLambdaMinusqCube := 0$$

$$Entry22TermLambdaMinusqCarre := -\frac{1}{5\, tinfty15}\left(h\left(-5\,\mu\, q\, tinfty14\, tinfty15 + 2\,\alpha 15\, h\, q + 5\, h\, \nu\, tinfty15 - 5\,\mu\, tinfty12\, tinfty15 + 10\,\rho\, tinfty15\right)\right)$$

$$Entry22TermLambdaMinusq := 0$$

$$Entry22TermLambdaInfty4 := 0$$

$$Entry22TermLambdaInfty3 := 0$$

$$Entry22TermLambdaInfty2 := 0$$

$$Entry22TermLambdaInfty1 := -\frac{4\, h\,(tinfty14\,\alpha 15 - 5\, c2\, tinfty15)}{5\, tinfty15}$$

$$Entry22TermLambdaInfty0 := -\frac{h\,(5\,\nu\, tinfty14\, tinfty15 + 2\,\alpha 15\, tinfty12 - 10\, c1\, tinfty15)}{5\, tinfty15}$$

$$0 \quad \textbf{(3)}$$

$$-tinfty14\,\lambda - tinfty12 + \frac{h}{\lambda - q} \quad \textbf{(3)}$$

Since the deformation operator is \hbar (alpha15\partial_{t_{\infty^{(1)},5} +alpha14\partial_{t_{\infty^{(1)},4}+alpha13*\partial_{t_{\infty^{(1)},3} +alpha12*\partial_{t_{\infty^{(1)},2}+alpha11*\partial_{t_{\infty^{(1)},1}) we get the evolutions of q and so consistency equations for the singular part at infinity

```
> L22OrderLambda2:=-residue(L[2,2]/lambda^3,lambda=infinity);
  L22OrderLambda1:=-residue(L[2,2]/lambda^2,lambda=infinity);
  L22OrderLambda0:=-residue(L[2,2]/lambda^1,lambda=infinity);
  Equation1:=factor(simplify(h*(alpha15*diff(L22OrderLambda1,
  tinfty15)+alpha14*diff(L22OrderLambda1,tinfty14)+alpha13*diff
  (L22OrderLambda1,tinfty13)+alpha12*diff(L22OrderLambda1,tinfty12)
  +alpha11*diff(L22OrderLambda1,tinfty11))-
  Entry22TermLambdaInfty1));
  Equation2:=factor(simplify(h*(alpha15*diff(L22OrderLambda0,
  tinfty15)+alpha14*diff(L22OrderLambda0,tinfty14)+alpha13*diff
  (L22OrderLambda0,tinfty13)+alpha12*diff(L22OrderLambda0,tinfty12)
  +alpha11*diff(L22OrderLambda0,tinfty11))-
  Entry22TermLambdaInfty0));
```

$$L22OrderLambda2 := 0$$
$$L22OrderLambda1 := -tinfty14$$
$$L22OrderLambda0 := -tinfty12$$
$$Equation1 := -\frac{h\,(5\,\alpha 14\,tinfty15 - 4\,tinfty14\,\alpha 15 + 20\,c2\,tinfty15)}{5\,tinfty15}$$
$$Equation2 := -\frac{h\,(-5\,\nu\,tinfty14\,tinfty15 + 5\,\alpha 12\,tinfty15 - 2\,\alpha 15\,tinfty12 + 10\,c1\,tinfty15)}{5\,tinfty15} \quad \textbf{(4)}$$

```
> CurlyLq:=factor(Entry22TermLambdaMinusqCarre/h);
  CurlyLqbis:=-mu*P1(q)-2*rho-h*nu-2/5/tinfty15*alpha15*h*q;
  factor(simplify(series(CurlyLq-CurlyLqbis,nu=0)));
```

$$CurlyLq := -\frac{1}{5\,tinfty15}\,(-5\,\mu\,q\,tinfty14\,tinfty15 + 2\,\alpha 15\,h\,q + 5\,h\,\nu\,tinfty15 - 5\,\mu\,tinfty12\,tinfty15 + 10\,\rho\,tinfty15)$$
$$CurlyLqbis := -\mu\,(-q\,tinfty14 - tinfty12) - 2\,\rho - h\,\nu - \frac{2\,\alpha 15\,h\,q}{5\,tinfty15}$$
$$0 \quad \textbf{(5)}$$

Let us now look at the compatibility equation for \mathcal{L}[L_{2,1}]

```
> Entry21:=simplify(CurlyL[2,1]):
  Entry21TermLambdaMinusqCube:=factor(residue(Entry21*(lambda-q)^2,
  lambda=q));
  Entry21TermLambdaMinusqCarre:=factor(residue(Entry21*(lambda-q),
  lambda=q));
  Entry21TermLambdaMinusq:=factor(residue(Entry21,lambda=q));
```

```
Entry21TermLambdaInfty6:=factor(-residue(Entry21/lambda^7,lambda=
infinity));
Entry21TermLambdaInfty5:=factor(-residue(Entry21/lambda^6,lambda=
infinity));
Entry21TermLambdaInfty4:=factor(-residue(Entry21/lambda^5,lambda=
infinity));
Entry21TermLambdaInfty3:=factor(-residue(Entry21/lambda^4,lambda=
infinity));
Entry21TermLambdaInfty2:=factor(-residue(Entry21/lambda^3,lambda=
infinity));
Entry21TermLambdaInfty1:=factor(-residue(Entry21/lambda^2,lambda=
infinity));
Entry21TermLambdaInfty0:=factor(-residue(Entry21/lambda,lambda=
infinity));

simplify( Entry21-(Entry21TermLambdaMinusqCube/(lambda-q)^3+
Entry21TermLambdaMinusqCarre/(lambda-q)^2+
Entry21TermLambdaMinusq/(lambda-q)
+Entry21TermLambdaInfty0+Entry21TermLambdaInfty1*lambda+
Entry21TermLambdaInfty2*lambda^2+Entry21TermLambdaInfty3*lambda^3
+Entry21TermLambdaInfty4*lambda^4+Entry21TermLambdaInfty5*
lambda^5+Entry21TermLambdaInfty6*lambda^6
) );
L[2,1];
```

$$Entry21TermLambdaMinusqCube := 3\, h^2\, (p\, \mu + \rho)$$

$$Entry21TermLambdaMinusqCarre := -\frac{1}{10\, tinfty15}\Big(h\, \big(5\, \mu\, q^3\, tinfty15^3 + 10\, q^2\, tinfty15^2\, tinfty13\, \mu - 5\, tinfty15\, q^2\, tinfty14^2\, \mu + 10\, \mu\, q\, tinfty11\, tinfty15^2 - 10\, \mu\, q\, tinfty12\, tinfty14\, tinfty15 + 5\, \mu\, q\, tinfty13^2\, tinfty15 - 4\, \alpha15\, q\, p\, h - 10\, tinfty15\, h\, p\, \nu + 10\, tinfty15\, q\, tinfty14\, \rho + 20\, tinfty15\, C\, \mu + 10\, tinfty15\, tinfty12\, \rho\big)\Big)$$

$$Entry21TermLambdaMinusq := -\frac{1}{20\, tinfty15}\Big(h\, \big(15\, \mu\, q^2\, tinfty15^3 + 20\, tinfty15^2\, q\, tinfty13\, \mu - 10\, tinfty15\, q\, tinfty14^2\, \mu + 40\, tinfty15\, q\, h\, c2 + 10\, tinfty15^2\, tinfty11\, \mu - 10\, tinfty15\, tinfty12\, tinfty14\, \mu + 5\, tinfty15\, tinfty13^2\, \mu + 8\, \alpha15\, p\, h + 20\, tinfty15\, h\, c1 + 20\, tinfty15\, tinfty14\, \rho\big)\Big)$$

$$Entry21TermLambdaInfty6 := 0$$
$$Entry21TermLambdaInfty5 := 0$$
$$Entry21TermLambdaInfty4 := 0$$

$$Entry21TermLambdaInfty3 := \frac{h\, \alpha15\, tinfty15}{2}$$

$$Entry21TermLambdaInfty2 :=$$

$$\frac{h\left(15\, tinfty15^3\, \nu + 16\, tinfty15\, \alpha 15\, tinfty13 - 8\, \alpha 15\, tinfty14^2 + 40\, tinfty15\, tinfty14\, c2\right)}{20\, tinfty15}$$

$$Entry21TermLambdaInfty1 := \frac{1}{20\, tinfty15}\left(h\left(5\, tinfty15^3\, \mu + 20\, tinfty15^2\, tinfty13\, \nu - 10\, tinfty15\, tinfty14^2\, \nu + 12\, tinfty15\, \alpha 15\, tinfty11 - 12\, \alpha 15\, tinfty14\, tinfty12 + 6\, \alpha 15\, tinfty13^2 + 20\, tinfty15\, tinfty14\, c1 + 40\, tinfty15\, tinfty12\, c2\right)\right)$$

$$Entry21TermLambdaInfty0 := \frac{1}{20\, tinfty15}\left(h\left(-5\, q\, tinfty15^3\, \mu + 10\, tinfty15^2\, tinfty11\, \nu - 10\, tinfty15\, tinfty12\, tinfty14\, \nu + 5\, tinfty15\, tinfty13^2\, \nu + 20\, tinfty15\, tinfty12\, c1 + 16\, \alpha 15\, C\right)\right)$$

$$0$$

$$-\left(-\frac{tinfty15\, tinfty11}{2} + \frac{tinfty14\, tinfty12}{2} - \frac{tinfty13^2}{4}\right)\lambda - \left(-\frac{tinfty15\, tinfty13}{2} + \frac{tinfty14^2}{4}\right)\lambda^2 + \frac{tinfty15^2\, \lambda^3}{4} + C - \frac{p\, h}{\lambda - q} \quad (6)$$

We now solve the compatibility equation for each term

```
> rho:=factor(solve(Entry21TermLambdaMinusqCube,rho));
  simplify(rho-(-p*mu));
  simplify(Entry21TermLambdaMinusqCube);
```

$$\rho := -p\, \mu$$
$$0$$
$$0 \quad (7)$$

```
> L21OrderLambda4:=-residue(L[2,1]/lambda^5,lambda=infinity);
  L21OrderLambda3:=-residue(L[2,1]/lambda^4,lambda=infinity);
  L21OrderLambda2:=-residue(L[2,1]/lambda^3,lambda=infinity);
  L21OrderLambda1:=-residue(L[2,1]/lambda^2,lambda=infinity);
  L21OrderLambda0:=-residue(L[2,1]/lambda^1,lambda=infinity);
  Equation3:=factor(simplify(h*(alpha15*diff(L21OrderLambda3,
  tinfty15)+alpha14*diff(L21OrderLambda3,tinfty14)+alpha13*diff
  (L21OrderLambda3,tinfty13)+alpha12*diff(L21OrderLambda3,tinfty12)
  +alpha11*diff(L21OrderLambda3,tinfty11))-
  Entry21TermLambdaInfty3));
  Equation4:=factor(simplify(h*(alpha15*diff(L21OrderLambda2,
  tinfty15)+alpha14*diff(L21OrderLambda2,tinfty14)+alpha13*diff
  (L21OrderLambda2,tinfty13)+alpha12*diff(L21OrderLambda2,tinfty12)
  +alpha11*diff(L21OrderLambda2,tinfty11))-
  Entry21TermLambdaInfty2));
  Equation5:=factor(simplify(h*(alpha15*diff(L21OrderLambda1,
  tinfty15)+alpha14*diff(L21OrderLambda1,tinfty14)+alpha13*diff
  (L21OrderLambda1,tinfty13)+alpha12*diff(L21OrderLambda1,tinfty12)
  +alpha11*diff(L21OrderLambda1,tinfty11))-
  Entry21TermLambdaInfty1));
```

```
Equation1:=factor(simplify(Equation1));
Equation2:=factor(simplify(Equation2));
```

$$L21OrderLambda4 := 0$$

$$L21OrderLambda3 := \frac{tinfty15^2}{4}$$

$$L21OrderLambda2 := \frac{tinfty15\ tinfty13}{2} - \frac{tinfty14^2}{4}$$

$$L21OrderLambda1 := \frac{tinfty15\ tinfty11}{2} - \frac{tinfty14\ tinfty12}{2} + \frac{tinfty13^2}{4}$$

$$L21OrderLambda0 := C$$

$$Equation3 := 0$$

$$Equation4 := \frac{1}{20\ tinfty15}\left(h\left(-15\ tinfty15^3\ \nu + 10\ \alpha 13\ tinfty15^2 - 10\ \alpha 14\ tinfty14\ tinfty15 - 6\ tinfty15\ \alpha 15\ tinfty13 + 8\ \alpha 15\ tinfty14^2 - 40\ tinfty15\ tinfty14\ c2\right)\right)$$

$$Equation5 := \frac{1}{20\ tinfty15}\left(h\left(-5\ tinfty15^3\ \mu - 20\ tinfty15^2\ tinfty13\ \nu + 10\ tinfty15\ tinfty14^2\ \nu + 10\ \alpha 1\ tinfty15^2 - 10\ \alpha 12\ tinfty14\ tinfty15 + 10\ \alpha 13\ tinfty13\ tinfty15 - 10\ \alpha 14\ tinfty12\ tinfty15 - 2\ tinfty15\ \alpha 15\ tinfty11 + 12\ \alpha 15\ tinfty14\ tinfty12 - 6\ \alpha 15\ tinfty13^2 - 20\ tinfty15\ tinfty14\ c1 - 40\ tinfty15\ tinfty12\ c2\right)\right)$$

$$Equation1 := -\frac{h\left(5\ \alpha 14\ tinfty15 - 4\ tinfty14\ \alpha 15 + 20\ c2\ tinfty15\right)}{5\ tinfty15}$$

$$Equation2 := -\frac{h\left(-5\ \nu\ tinfty14\ tinfty15 + 5\ \alpha 12\ tinfty15 - 2\ \alpha 15\ tinfty12 + 10\ c1\ tinfty15\right)}{5\ tinfty15} \quad \textbf{(8)}$$

```
> solve({Equation1,Equation2,Equation4,Equation5},{c2,c1,mu,nu});
```

$$\Big\{ c1 = -\frac{1}{30\ tinfty15^2}\left(15\ \alpha 12\ tinfty15^2 - 10\ \alpha 13\ tinfty14\ tinfty15 - 6\ \alpha 15\ tinfty12\ tinfty15 + 6\ \alpha 15\ tinfty13\ tinfty14\right), c2 = -\frac{5\ \alpha 14\ tinfty15 - 4\ tinfty14\ \alpha 15}{20\ tinfty15}, \mu = \frac{1}{15\ tinfty15^3}\left(2\left(15\ \alpha 11\ tinfty15^2 - 5\ \alpha 13\ tinfty13\ tinfty15 - 3\ tinfty15\ \alpha 15\ tinfty11 + 3\ \alpha 15\ tinfty13^2\right)\right), \nu = \frac{2\left(5\ \alpha 13\ tinfty15 - 3\ \alpha 15\ tinfty13\right)}{15\ tinfty15^2} \Big\} \quad \textbf{(9)}$$

```
> c1bis := (6*alpha15*tinfty12*tinfty15-6*alpha15*tinfty13*
tinfty14+10*alpha13*tinfty14*tinfty15-15*alpha12*tinfty15^2)/(30*
tinfty15^2);
c2bis:= (4*alpha15*tinfty14-5*alpha14*tinfty15)/(20*tinfty15);
mubis:=(2*(-3*alpha15*tinfty11*tinfty15+3*alpha15*tinfty13^2
```

```
-5*alpha13*tinfty13*tinfty15+15*alpha11*tinfty15^2))/(15*
tinfty15^3);
nubis:=-(2*(3*alpha15*tinfty13-5*alpha13*tinfty15))/(15*
tinfty15^2);
simplify(c1bis-c1Asymp);
simplify(c2bis-c2Asymp);
simplify(mubis-muAsymp);
simplify(nubis-nuAsymp);
```

$$c1bis := \frac{1}{30\, tinfty15^2}\left(-15\,\alpha 12\, tinfty15^2 + 10\,\alpha 13\, tinfty14\, tinfty15 + 6\,\alpha 15\, tinfty12\, tinfty15 - 6\,\alpha 15\, tinfty13\, tinfty14\right)$$

$$c2bis := \frac{-5\,\alpha 14\, tinfty15 + 4\, tinfty14\,\alpha 15}{20\, tinfty15}$$

$$mubis := \frac{2\left(15\,\alpha 11\, tinfty15^2 - 5\,\alpha 13\, tinfty13\, tinfty15 - 3\, tinfty15\,\alpha 15\, tinfty11 + 3\,\alpha 15\, tinfty13^2\right)}{15\, tinfty15^3}$$

$$nubis := -\frac{2\left(-5\,\alpha 13\, tinfty15 + 3\,\alpha 15\, tinfty13\right)}{15\, tinfty15^2}$$

$$0$$
$$0$$
$$0$$
$$0 \qquad (10)$$

```
> CurlyLpfunction:=unapply(-Entry21TermLambdaMinusq/h,C):
> Equation7:=simplify(Entry21TermLambdaMinusqCarre-(-p*h*CurlyLq)):
> Csol:=solve(Equation7,C):
  Csolbis:=p^2- P1(q)*p+P2(q)-Pinfty02:
  factor(series(Csol-Csolbis,p=0));
```

$$0 \qquad (11)$$

```
> CurlyLp:=factor(simplify(CurlyLpfunction(Csol))):
  CurlyLpbis:=mu*(p*diff(P1(q),q) -diff(P2(q),q))
  +h*2/5/tinfty15*alpha15*p
  +2*h*c2*q+h*c1:
  factor(series(CurlyLp-CurlyLpbis,q=0));
```

$$0 \qquad (12)$$

Conclusion: We have verified that the formulas for c1,c2,mu, are compatible with the theory.
Moreover, we get the evolutions of the Darboux coordinates:
L[q]=2*mu*p-mu*P1(q)-h*nu-2/5/tinfty15*alpha15*h*q;
L[p] = mu*(p*diff(P1(q),q) -diff(P2(q),q))+h*2/5/tinfty15*alpha15*p+2*h*c2*q+h*c1

```
> CurlyLqbis:=2*p*mu-mu*P1(q)-h*nu-2/5/tinfty15*alpha15*h*q:
  factor(simplify(series(CurlyLq-CurlyLqbis,q=0)))
```

$$0 \qquad (13)$$

```
> Hamiltonian:= mu*(p^2+P2(q)-p*P1(q))-h*nu*p-2/5/tinfty15*h*
  alpha15*p*q -h*c2*q^2-h*c1*q:
```

```
simplify(CurlyLp-(-diff(Hamiltonian,q)));
simplify(CurlyLq-(diff(Hamiltonian,p)));
```

$$0$$
$$0 \qquad (14)$$

```
> C:=Csolbis:
  c1:=c1bis:
  c2:=c2bis:
  mu:=mubis:
  nu:=nubis:
```

Decomposition of the tangent space into trivial and non-trivial directions

```
> T1:=tinfty13/(2*rinfty-3)*(1/2*tinfty15)^(-(2*rinfty-3)/(2*
  rinfty-1));
  T2:=(1/2*tinfty15)^(2/(2*rinfty-1));
  tau:=(1/2*tinfty15)^((2*rinfty-3-2*1)/(2*rinfty-1))*1/2*tinfty11;
  checkq:=T2*q+T1;
  checkp:=T2^(-1)*(p-1/2*P1(q));

  tau1theo:=unapply( (1/2*tinfty15)^(-(2*rinfty-5)/(2*rinfty-1))*
  1/2*tinfty11- (2*rinfty-5)/(2*rinfty-3)/2*(1/2*tinfty13)^2*(1/2*
  tinfty15)^(-2*(2*rinfty-3)/(2*rinfty-1)), tinfty11);
  tinfty11theo:=unapply( 2*T2^(1/2)*(tau1+T1^2*(2*rinfty-3)*(2*
  rinfty-5)/2^2/(2!)), tau1);
  simplify(tinfty11theo(tau1theo(t)));
```

$$T1 := \frac{tinfty13\, 2^{3/5}}{3\, tinfty15^{3/5}}$$

$$T2 := \frac{2^{3/5}\, tinfty15^{2/5}}{2}$$

$$\tau := \frac{2^{4/5}\, tinfty15^{1/5}\, tinfty11}{4}$$

$$checkq := \frac{2^{3/5}\, tinfty15^{2/5}\, q}{2} + \frac{tinfty13\, 2^{3/5}}{3\, tinfty15^{3/5}}$$

$$checkp := \frac{2^{2/5}\left(p + \frac{tinfty14\, q}{2} + \frac{tinfty12}{2}\right)}{tinfty15^{2/5}}$$

$$tau1theo := tinfty11 \mapsto \frac{2^{1/5}\, tinfty11}{2\, tinfty15^{1/5}} - \frac{tinfty13^{2}\, 2^{1/5}}{12\, tinfty15^{6/5}}$$

$$tinfty11theo := \tau 1 \mapsto \sqrt{2}\, \sqrt{2^{3/5}\, tinfty15^{2/5}} \left(\tau 1 + \frac{tinfty13^{2}\, 2^{1/5}}{12\, tinfty15^{6/5}}\right)$$

$$t \qquad (15)$$

```
> LuMinus1T1:=h/2*(5*tinfty15*diff(T1,tinfty15)+4*tinfty14*diff(T1,
  tinfty14)+3*tinfty13*diff(T1,tinfty13)+2*tinfty12*diff(T1,
  tinfty12)+1*tinfty11*diff(T1,tinfty11));
```

```
LuMinus1T2:=h/2*(5*tinfty15*diff(T2,tinfty15)+4*tinfty14*diff(T2,
tinfty14)+3*tinfty13*diff(T2,tinfty13)+2*tinfty12*diff(T2,
tinfty12)+1*tinfty11*diff(T2,tinfty11));
simplify(LuMinus1T2-h*T2);
LuMinus1P1:=unapply( h/2*(5*tinfty15*diff(P1(lambda),tinfty15)+4*
tinfty14*diff(P1(lambda),tinfty14)+3*tinfty13*diff(P1(lambda),
tinfty13)+2*tinfty12*diff(P1(lambda),tinfty12)+1*tinfty11*diff(P1
(lambda),tinfty11)),lambda);
CurlyLqfunction:=unapply(CurlyLqbis,alpha15,
alpha14,alpha13,alpha12,alpha11):
CurlyLpfunction:=unapply(CurlyLpbis,alpha15,alpha14,alpha13,
alpha12,alpha11):
LuMinus1q:=CurlyLqfunction(5*tinfty15/2,4*tinfty14/2,3*
tinfty13/2,2*tinfty12/2,1*tinfty11/2);
LuMinus1p:=CurlyLpfunction(5*tinfty15/2,4*tinfty14/2,3*
tinfty13/2,2*tinfty12/2,1*tinfty11/2);

LuMinus1checkq:=simplify(h/2*(5*tinfty15*diff(checkq,tinfty15)+4*
tinfty14*diff(checkq,tinfty14)+3*tinfty13*diff(checkq,tinfty13)
+2*tinfty12*diff(checkq,tinfty12)+1*tinfty11*diff(checkq,
tinfty11)) +diff(checkq,q)*LuMinus1q+diff(checkq,p)*LuMinus1p);
LuMinus1checkp:=simplify(h/2*(5*tinfty15*diff(checkp,tinfty15)+4*
tinfty14*diff(checkp,tinfty14)+3*tinfty13*diff(checkp,tinfty13)
+2*tinfty12*diff(checkp,tinfty12)+1*tinfty11*diff(checkp,
tinfty11)) +diff(checkp,q)*LuMinus1q+diff(checkp,p)*LuMinus1p);
```

$$LuMinus1T1 := 0$$

$$LuMinus1T2 := \frac{h\, 2^{3/5}\, tinfty15^{2/5}}{2}$$

$$0$$

$$LuMinus1P1 := \lambda \mapsto \frac{h\,(-4\,\lambda\, tinfty14 - 2\, tinfty12)}{2}$$

$$LuMinus1q := -h\, q$$

$$LuMinus1p := p\, h$$

$$LuMinus1checkq := 0$$

$$LuMinus1checkp := 0 \qquad (16)$$

```
> Lu0T1:=h/2*(3*tinfty15*diff(T1,tinfty13)+2*tinfty14*diff(T1,
tinfty12)+1*tinfty13*diff(T1,tinfty11));
simplify(Lu0T1-h*T2);
Lu0T2:=h/2*(3*tinfty15*diff(T2,tinfty13)+2*tinfty14*diff(T2,
tinfty12)+1*tinfty13*diff(T2,tinfty11));
simplify(Lu0T2);
Lu0P1:=unapply( h/2*(3*tinfty15*diff(P1(lambda),tinfty13)+2*
tinfty14*diff(P1(lambda),tinfty12)+1*tinfty13*diff(P1(lambda),
```

```
tinfty11)),lambda);

Lu0q:=CurlyLqfunction(0,0,3*tinfty15/2,2*tinfty14/2,1*tinfty13/2)
;
Lu0p:=CurlyLpfunction(0,0,3*tinfty15/2,2*tinfty14/2,1*tinfty13/2)
;

Lu0checkq:=simplify(h/2*(3*tinfty15*diff(checkq,tinfty13)+2*
tinfty14*diff(checkq,tinfty12)+1*tinfty13*diff(checkq,tinfty11))
+diff(checkq,q)*Lu0q+diff(checkq,p)*Lu0p);
Lu0checkp:=simplify(h/2*(3*tinfty15*diff(checkp,tinfty13)+2*
tinfty14*diff(checkp,tinfty12)+1*tinfty13*diff(checkp,tinfty11))
+diff(checkp,q)*Lu0q+diff(checkp,p)*Lu0p);
```

$$Lu0T1 := \frac{h\, 2^{3/5}\, tinfty15^{2/5}}{2}$$

$$0$$

$$Lu0T2 := 0$$

$$0$$

$$Lu0P1 := \lambda \mapsto -h\, tinfty14$$

$$Lu0q := -h$$

$$Lu0p := 0$$

$$Lu0checkq := 0$$

$$Lu0checkp := 0 \qquad \mathbf{(17)}$$

> We define tdp:=p-P1(q)/2

```
> p:=tdp+P1(q)/2:
CurlyLtdp:=simplify( CurlyLp-dP1dlambda(q)/2*CurlyLq
- 1/2*h*(alpha15*diff(P1(q),tinfty15)+alpha14*diff(P1(q),
tinfty14)+alpha13*diff(P1(q),tinfty13)+alpha12*diff(P1(q),
tinfty12)+alpha11*diff(P1(q),tinfty11)) ):
> CurlyLtdpbis:=mu*(diff(P1(q)^2/4-P2(q),q))+h*2/5/tinfty15*
alpha15*tdp:
factor(series(CurlyLtdp-CurlyLtdpbis,q=0));
```

$$0 \qquad \mathbf{(18)}$$

```
> simplify(CurlyLq- ( 2*mu*tdp-h*nu-2/5/tinfty15*alpha15*h*q));
```

$$0 \qquad \mathbf{(19)}$$

We get that \mathcal{L}[q]=2*mu*tdp-h*nu-h*2/5/tinfty15*alpha15*q;

mathcal{L}[tdp] = mu*(diff(P1(q)^2/4-P2(q),q))+h*2/5/tinfty15*alpha15*tdp

Hamiltonian= mu*tdp^2 -h*2/5/tinfty15*alpha15*q*tdp -mu*(P1(q)^2/4-P2(q)) -h*nu*tdp

```
> mufunction:=unapply(mubis,alpha15,
alpha14,alpha13,alpha12,alpha11);
nufunction:=unapply(nu,alpha15,alpha14,alpha13,alpha12,alpha11);
c2function:=unapply(c2,alpha15,alpha14,alpha13,alpha12,alpha11);
```

```
c1function:=unapply(c1,alpha15,alpha14,alpha13,alpha12,alpha11);
CurlyLqfunction:=unapply(CurlyLq,alpha15,alpha14,alpha13,alpha12,
alpha11,q,tdp):
CurlyLtdpfunction:=unapply(CurlyLtdp,alpha15,alpha14,alpha13,
alpha12,alpha11,q,tdp):

CurlyLcheckqfunction:=unapply( h*(diff(T2,tinfty15)*alpha15+diff
(T2,tinfty14)*alpha14+diff(T2,tinfty13)*alpha13+diff(T2,tinfty12)
*alpha12+diff(T2,tinfty11)*alpha11)*qfunction+ T2*CurlyLqfunction
(alpha15,alpha14,alpha13,alpha12,alpha11,qfunction,tdpfunction)+
h*(diff(T1,tinfty15)*alpha15+diff(T1,tinfty14)*alpha14+diff(T1,
tinfty13)*alpha13+diff(T1,tinfty12)*alpha12+diff(T1,tinfty11)*
alpha11),
alpha15,alpha14,alpha13,alpha12,alpha11):

CurlyLcheckpfunction:=unapply(-h/T2^2*(diff(T2,tinfty15)*alpha15+
diff(T2,tinfty14)*alpha14+diff(T2,tinfty13)*alpha13+diff(T2,
tinfty12)*alpha12+diff(T2,tinfty11)*alpha11)*tdpfunction+T2^(-1)*
CurlyLtdpfunction(alpha15,alpha14,alpha13,alpha12,alpha11,
qfunction,tdpfunction),
alpha15,alpha14,alpha13,alpha12,alpha11):
```

$$mufunction := (\alpha l5, \alpha l4, \alpha l3, \alpha l2, \alpha l1) \mapsto \frac{1}{15\, tinfty15^3}\left(2\left(15\, \alpha l1\, tinfty15^2 - 5\, \alpha l3\, tinfty13\, tinfty15 - 3\, \alpha l5\, tinfty11\, tinfty15 + 3\, \alpha l5\, tinfty13^2\right)\right)$$

$$nufunction := (\alpha l5, \alpha l4, \alpha l3, \alpha l2, \alpha l1) \mapsto -\frac{2\left(-5\, \alpha l3\, tinfty15 + 3\, \alpha l5\, tinfty13\right)}{15\, tinfty15^2}$$

$$c2function := (\alpha l5, \alpha l4, \alpha l3, \alpha l2, \alpha l1) \mapsto \frac{-5\, \alpha l4\, tinfty15 + 4\, \alpha l5\, tinfty14}{20\, tinfty15}$$

$$c1function := (\alpha l5, \alpha l4, \alpha l3, \alpha l2, \alpha l1) \mapsto \frac{1}{30\, tinfty15^2}\left(-15\, \alpha l2\, tinfty15^2 + 10\, \alpha l3\, tinfty14\, tinfty15 + 6\, \alpha l5\, tinfty12\, tinfty15 - 6\, \alpha l5\, tinfty13\, tinfty14\right) \quad (20)$$

Evolution in direction w2

```
> mufunction(0,1,0,0,0);
nufunction(0,1,0,0,0);
c2function(0,1,0,0,0);
c1function(0,1,0,0,0);
CurlyLqfunction(0,1,0,0,0,q,tdp);
CurlyLtdpfunction(0,1,0,0,0,q,tdp);
simplify(CurlyLcheckqfunction(0,1,0,0,0));
simplify(CurlyLcheckpfunction(0,1,0,0,0));
```

$$\begin{bmatrix} 0 \\ 0 \\ -\frac{1}{4} \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \quad (21)$$

Evolution in direction w1

```
> mufunction(0,0,0,1,0);
  nufunction(0,0,0,1,0);
  c2function(0,0,0,1,0);
  c1function(0,0,0,1,0);
  CurlyLqfunction(0,0,0,1,0,q,tdp);
  CurlyLtdpfunction(0,0,0,1,0,q,tdp);
  simplify(CurlyLcheckqfunction(0,0,0,1,0));
  simplify(CurlyLcheckpfunction(0,0,0,1,0));
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ -\frac{1}{2} \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \quad (22)$$

Evolution in direction u0

```
> mufunction(0,0,1/2*3*tinfty15,1/2*2*tinfty14,1/2*tinfty13);
  nufunction(0,0,1/2*3*tinfty15,1/2*2*tinfty14,1/2*tinfty13);
  c2function(0,0,1/2*3*tinfty15,1/2*2*tinfty14,1/2*tinfty13);
  c1function(0,0,1/2*3*tinfty15,1/2*2*tinfty14,1/2*tinfty13);
  CurlyLqfunction(0,0,1/2*3*tinfty15,1/2*2*tinfty14,1/2*tinfty13,q,
  tdp);
  simplify(CurlyLtdpfunction(0,0,1/2*3*tinfty15,1/2*2*
  tinfty14,1/2*tinfty13,q,tdp));
  simplify(CurlyLcheckqfunction(0,0,1/2*3*tinfty15,1/2*2*tinfty14,
  1/2*tinfty13));
  simplify(CurlyLcheckpfunction(0,0,1/2*3*tinfty15,1/2*2*tinfty14,
  1/2*tinfty13));
```

$$\begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \\ -h \\ 0 \\ 0 \end{bmatrix}$$

$$0 \quad (23)$$

Evolution in direction u_{-1}

```
> mufunction(1/2*5*tinfty15,1/2*4*tinfty14,1/2*3*tinfty13,1/2*2*
  tinfty12,1/2*tinfty11);
  nufunction(1/2*5*tinfty15,1/2*4*tinfty14,1/2*3*tinfty13,1/2*2*
  tinfty12,1/2*tinfty11);
  c2function(1/2*5*tinfty15,1/2*4*tinfty14,1/2*3*tinfty13,1/2*2*
  tinfty12,1/2*tinfty11);
  c1function(1/2*5*tinfty15,1/2*4*tinfty14,1/2*3*tinfty13,1/2*2*
  tinfty12,1/2*tinfty11);
  CurlyLqfunction(1/2*5*tinfty15,1/2*4*tinfty14,1/2*3*tinfty13,1/2*
  2*tinfty12,1/2*tinfty11,q,tdp);
  simplify(CurlyLtdpfunction(1/2*5*tinfty15,1/2*4*tinfty14,1/2*3*
  tinfty13,1/2*2*tinfty12,1/2*tinfty11,q,tdp));
  simplify(CurlyLcheckqfunction(1/2*5*tinfty15,1/2*4*tinfty14,1/2*
  3*tinfty13,1/2*2*tinfty12,1/2*tinfty11));
  simplify(CurlyLcheckpfunction(1/2*5*tinfty15,1/2*4*tinfty14,1/2*
  3*tinfty13,1/2*2*tinfty12,1/2*tinfty11));
```

$$0$$
$$0$$
$$0$$
$$0$$
$$-h\,q$$
$$h\,tdp$$
$$0$$
$$0 \quad (24)$$